# Vector fields and complex integral

In this note, we will use Mathematica to

- Visualize ideal flows using holomorphic functions.
- Approximate complex integrals numerically.

## 1 Ideal flows, potential function and stream function

Recall that a complex function $f : D \subset \mathbb{C} \to \mathbb{C}$ can be viewed as a geometric transformation: each point in $D$ (a region on the plane) is mapped to a point on the plane. We now consider another geometric meaning of complex functions: *vector field*. Let us write $f$ in standard form:

$$f(z) = u(x, y) + iv(x, y)$$

where $u$ and $v$ are real-valued functions. Function $f$ places at each point $(x, y)$ a vector $(u, v) = (u(x, y), v(x, y))$. In Mathematica, one can use the command **VectorPlot** to visualize a vector field. The basic syntax is:

```
VectorPLot[{u[x,y],v[x,y]},{x,a,b},{y,c,d}]
```

For example, the vector field $(u, v) = (x + y, \sin(y))$ is visualized as (Figure 1):

```
VectorPlot[{x + y, Sin[y]}, {x, -2, 2}, {y, -2, 2}]
```




Figure 1

One can observe from Figure 1 that the command VectorPlot does not show precisely the vector field $(x + y, \sin(y))$. Indeed, if that were the case then the vector at position $(x, y) = (2, 2)$ would be $(2 + 2, \sin 2) \approx (4, 0.91)$, which is quite a long vector. Instead, the command VectorPlot shows a scaled version of the vector field, which is easier to see and still gives good perception of direction and magnitude.

A vector field $F(x, y) = (u(x, y), v(x, y))$ is said to be *incompressible* if it is divergence-free, i.e.

$$\operatorname{div} F = u_x + v_y = 0 \quad \forall (x, y).$$

$F$ is said to be *irrotational* if it is curl-free, i.e.

$$\operatorname{curl} F = v_x - u_y = 0 \quad \forall (x, y).$$

The vector field in Figure 1 is neither incompressible nor irrotational. Indeed, the vector field seems to have a vortex at the origin (i.e. near the origin the vector field rotates about the origin). Also, the origin seems to be a source point (where vectors in the vector field are generated), making $F$ not incompressible. One can compute $\operatorname{div}F(0,0)$ and $\operatorname{curl}F(0,0)$ to verify this observation. An incompressible irrotational vector field is said to be an *ideal flow*. There is an interesting connection between holomorphic functions and ideal flows as follows.

**Let $f = u + iv$ be a holomorphic function in region $D \subset \mathbb{C}$. Then vector field $P = (u, -v)$ is an ideal flow in $D$ (Problem 1, Homework 7).** It is called *Polya vector field* associated with $f$. The converse is also true: every ideal flow corresponds to a holomorphic complex function.

If $f_1(z) = z$ then $P(x, y) = (x, -y)$. If $f(z) = z^2$ then $P(x, y) = (x^2 - y^2, -2xy)$. In Mathematica,

```
f[z_] := z^2
VectorPlot[{Re[f[x + I*y]], -Im[f[x + I*y]]}, {x, -2, 2}, {y, -2, 2},
       VectorStyle -> Blue]
```




(a) Ideal flow with $f(z) = z$

(b) Ideal flow with $f(z) = z^2$

Figure 2

Each vector field in Figure 2 is the velocity field of a fluid flow. Streamlines are the trajectories of fluid particles. Thus, the vector field at each point on a streamline is tangent to the streamline. One can draw streamlines with Mathematica (Figure 3):

```
StreamPlot[{Re[f[x + I*y]], -Im[f[x + I*y]]}, {x, -2, 2}, {y, -2, 2},
 StreamStyle -> Orange]
```

A natural question is: *how to find the equation of the streamlines?* There is a connection between streamlines and antiderivative of $f$ as follows.



(a) Streamlines of flow with $f(z) = z$ (b) Streamlines of flow with $f(z) = z^2$

Figure 3

**Let $\Phi(z) = \phi(x, y) + i\psi(x, y)$ be an antiderivative of $f$. Then the level sets of $\psi$ are the streamlines of the ideal flow associated with $f$. (Problem 2, Homework 7).** For this reason, $\psi$ is called the *stream function* of $f$ (or of the ideal flow associated with $f$). Function $\phi$ is called the *potential function* of the flow. Function $\Phi$ is called the *complex potential* of the flow. Thus, the problem of finding streamlines of a flow becomes the problem of finding antiderivative of a complex function.

Let us use $f(z) = z$ as an example. An antiderivative of $f$ is $\Phi(z) = \frac{1}{2}z^2 = \frac{1}{2}(x^2 - y^2) + ixy$. The stream function of the flow is $\psi(x, y) = xy$. The streamlines of the flow are the level sets of $\psi$, which are the sets of $(x, y)$ such that

$$\psi(x, y) = xy = C \qquad \text{(const)}.$$

This explains the hyperbola curves in Figure 3a. The streamline corresponding to $C = 0$ is special. It consists of the $x$-axis and $y$-axis. They divide the plane into four quadrants. A fluid particle that starts in a quadrant will never escape that quadrant (according to the picture). In this way, the two axes acts like physical boundaries of the fluid flow. One can view the flow in Figure 3a as a flow in the corner of two infinite walls (Figure 4). In most situations, it is not easy to solve for $(x, y)$ from the equation $\psi(x, y) = C$. However, one can use the command **ContourPlot** to plot the solution set of an equation.

```
f[z_] := z
p = StreamPlot[{Re[f[x + I*y]], -Im[f[x + I*y]]}, {x, -2, 2}, {y, -2,
   2}, StreamStyle -> Orange]
q = ContourPlot[{x*y == 0}, {x, -2, 2}, {y, -2, 2}]
Show[p, q]
```

## 2 Complex integral

Let us consider the integral $\int_\gamma f(z)dz$ where $f(z) = e^z$ and $\gamma$ is the curve $y = \sin x$ where $x \in [0, 3\pi]$. This integral can be computed using the Fundamental Theorem of Calculus, knowing that an antiderivative of $f(z)$ is $e^z$. However, for the sake of generality, let us try to compute the integral



Figure 4

using the general formula

$$\int_\gamma f(z)dz = \int_a^b f(\gamma(t))\gamma'(t)dt.$$

We can parametrize $\gamma$ as $\gamma(t) = x(t) + iy(t) = t + i\sin t$ where $0 \leq t \leq 3\pi$. Then

$$\int_\gamma f(z)dz = \int_0^{3\pi} e^{t+i\sin t}(1 + i\cos t)dt.$$

We can try the command **Integrate** on Mathematica to compute this integral:

```
f[z_] := Exp[z]
gamma[t_] := t + I*Sin[t]
Dgamma[t_] := 1 + I*Cos[t]
Integrate[f[gamma[t]]*Dgamma[t], {t, 0, 3*Pi}]
```

Mathematica takes quite some time to give an answer, which is $e^{3\pi} - 1$. It takes less time to compute an approximate numerical value of the integral:

```
NIntegrate[f[gamma[t]]*Dgamma[t], {t, 0, 3*Pi}]
```

This command is useful when one wants to double check the result of $\int_\gamma f(z)dz$ obtained from other methods.